Problem:
Score : 400 points

Problem Statement

Given is a rooted tree with N vertices numbered 1 to N.
The root is Vertex 1, and the i-th edge (1 \leq i \leq N - 1) connects Vertex a_i and b_i.

Each of the vertices has a counter installed. Initially, the counters on all the vertices have the value 0.

Now, the following Q operations will be performed:

Operation j (1 \leq j \leq Q): Increment by x_j the counter on every vertex contained in the subtree rooted at Vertex p_j.

Find the value of the counter on each vertex after all operations.

Constraints

2 \leq N \leq 2 \times 10^5

1 \leq Q \leq 2 \times 10^5

1 \leq a_i < b_i \leq N

1 \leq p_j \leq N

1 \leq x_j \leq 10^4

The given graph is a tree.

All values in input are integers.

Input

Input is given from Standard Input in the following format:

N Q
a_1 b_1
:
a_{N-1} b_{N-1}
p_1 x_1
:
p_Q x_Q

Output

Print the values of the counters on Vertex 1, 2, \ldots, N after all operations, in this order, with spaces in between.

Sample Input 1

4 3
1 2
2 3
2 4
2 10
1 100
3 1

Sample Output 1

100 110 111 110

The tree in this input is as follows:

Each operation changes the values of the counters on the vertices as follows:

Operation 1: Increment by 10 the counter on every vertex contained in the subtree rooted at Vertex 2, that is, Vertex 2, 3, 4. The values of the counters on Vertex 1, 2, 3, 4 are now 0, 10, 10, 10, respectively.

Operation 2: Increment by 100 the counter on every vertex contained in the subtree rooted at Vertex 1, that is, Vertex 1, 2, 3, 4. The values of the counters on Vertex 1, 2, 3, 4 are now 100, 110, 110, 110, respectively.

Operation 3: Increment by 1 the counter on every vertex contained in the subtree rooted at Vertex 3, that is, Vertex 3. The values of the counters on Vertex 1, 2, 3, 4 are now 100, 110, 111, 110, respectively.

Sample Input 2

6 2
1 2
1 3
2 4
3 6
2 5
1 10
1 10

Sample Output 2

20 20 20 20 20 20

Solution in Pascal (FPC 2.6.2) (judge verdict: Accepted)

var
  a, h, f, w, id, hs, sum: array [0..200000] of longint;
  u, v: array [0..400000] of longint;
  s, t: array [1..800000] of longint;
  n, q, i, x, y, num: longint;
procedure swap(var x, y: longint); inline;
var
  t: longint;
begin
  t := x;
  x := y;
  y := t;
end;
procedure add(x, y: longint); inline;
begin
  inc(num);
  u[num] := h[x];
  v[num] := y;
  h[x] := num;
end;
procedure pushup(x: longint); inline;
begin
  s[x] := s[x shl 1] + s[x shl 1 or 1];
end;
procedure pushdown(x, l, r: longint); inline;
var
  m: longint;
begin
  m := (l + r) shr 1;
  inc(s[x shl 1], t[x] * (m - l + 1));
  inc(s[x shl 1 or 1], t[x] * (r - m));
  inc(t[x shl 1], t[x]);
  inc(t[x shl 1 or 1], t[x]);
  t[x] := 0;
end;
procedure build(x, l, r: longint);
var
  m: longint;
begin
  if l = r then
  begin
    s[x] := w[l];
    exit;
  end;
  m := (l + r) shr 1;
  build(x shl 1, l, m);
  build(x shl 1 or 1, m + 1, r);
  pushup(x);
end;
procedure update(x, l, r, tl, tr, k: longint);
var
  m: longint;
begin
  if (l >= tl) and (r <= tr) then
  begin
    inc(s[x], k * (r - l + 1));
    inc(t[x], k);
    exit;
  end;
  pushdown(x, l, r);
  m := (l + r) shr 1;
  if tl <= m then
    update(x shl 1, l, m, tl, tr, k);
  if tr > m then
    update(x shl 1 or 1, m + 1, r, tl, tr, k);
  pushup(x);
end;
function query(x, l, r, t: longint): longint;
var
  m: longint;
begin
  if l = r then  
    exit(s[x]);
  pushdown(x, l, r);
  m := (l + r) shr 1;
  if t <= m then
    exit(query(x shl 1, l, m, t))
  else
    exit(query(x shl 1 or 1, m + 1, r, t));
end;
procedure dfs1(x, y, z: longint);
var
  i, max: longint;
begin
  f[x] := y;
  sum[x] := 1;
  hs[x] := 0;
  max := 0;
  i := h[x];
  while i > 0 do
  begin
    if v[i] <> y then
    begin
      dfs1(v[i], x, z + 1);
      if sum[v[i]] > max then
      begin
        max := sum[v[i]];
        hs[x] := v[i];
      end;
      inc(sum[x], sum[v[i]]);
    end;
    i := u[i];
  end;
end;
procedure dfs2(x, y: longint);
var
  i: longint;
begin
  inc(num);
  id[x] := num;
  w[id[x]] := a[x];
  if hs[x] = 0 then
    exit;
  dfs2(hs[x], y);
  i := h[x];
  while i > 0 do
  begin
    if (v[i] <> f[x]) and (v[i] <> hs[x]) then
      dfs2(v[i], v[i]);
    i := u[i];
  end;
end;
procedure updatesubtree(x, k: longint); inline;
begin
  update(1, 1, n, id[x], id[x] + sum[x] - 1, k);
end;
begin
  read(n, q);
  num := 0;
  for i := 1 to n - 1 do
  begin
    read(x, y);
    add(x, y);
    add(y, x);
  end;
  num := 0;
  dfs1(1, 0, 0);
  dfs2(1, 1);
  build(1, 1, n);
  for i := 1 to q do
  begin
    read(x, y);
    updatesubtree(x, y);
  end;
  for i := 1 to n do
    write(query(1, 1, n, id[i]), ' ');
end.